Short Introduction to Reliability Engineering and PROC RELIABILITY to Non-Engineers

Danny Rithy, California Polytechnic State University, San Luis Obispo, CA

ABSTRACT

Reliability engineering specializes how often a product or system fails under stated conditions over time. In the modern world, it is important for a product or system maintains for a long time. Because technology is well-developed these days, some systems will eventually fail. Mathematical and statistical methods are useful for quantifying and for analyzing reliability data. However, the most important priority of reliability engineering is to apply engineering knowledge to prevent the likelihood of failures. This paper introduces the idea of reliability engineering to non-engineers as well as PROC RELIABILITY that demonstrates some applications of reliability data.

INTRODUCTION

Modern technology is a very powerful tool to make life simpler. Consumers rely on machines they need to use every day to meet their needs. When consumers buy a product, they're expecting the system to work for a very long period of time. Nobody wants to own a faulty product where it fails repeatedly over time. This attention goes to the manufacturing companies. How do manufacturing companies make their products more competitive and trustworthy? How do they identify and correct the failures that do occur? How can we reduce the frequency of failures? The process behind this is based on reliability engineering.

Reliability engineering studies the probability that a system will function without failure under stated conditions over time. The main objective of reliability engineering is to optimize the reliability of the systems in the long run while minimizing cost. Reliability engineering differs from quality control where it evaluates the effectiveness and efficiency of the system while it meets specific requirements.

In general, we want to own products that are long-lasting. So we want to have higher reliability to meet customer requirements. From a customer's perspective, we want to own things that have higher durability. From a company's perspective, we want to maintain the reputation with minimum complaints from customers. The goal of any company is to make money. This means inefficient manufacturing products can cause failures and possibly safety hazards.

Products manufactured for human use or operation are typically designed with safety in mind. To prevent serious injury or death during their operation or use, products often have some sort of implanted safety mechanism in place. For example, should a smartphone start to overheat, they are often designed to shut down to prevent injury to its operator and damage to the device itself. Another example is when a car crashes, an airbag is immediately deployed to protect its occupants. Reliability engineering determines how dependable or reliable a manufactured product or system is under various operating conditions and uses. This process, known as Failure Mode Effects Analysis evaluates possible failure scenarios along with their effect on the user or occupant. In addition, reliability engineers have to rate each effect on the level of likelihood, seriousness, and danger.

Patrick O'Connor specifies the objectives of reliability engineering as:

1. Apply engineering knowledges and techniques to find ways to reduce the frequency of failures.

2. Identify the cause and correct the failures they do occur.

3. Determine ways to deal with failures that do occur.

4. Use analytical methods to estimate reliability of the design or analyze reliability data.

This paper should give you an idea how reliability engineering works in the industry and how engineers use SAS® and PROC RELIABILITY to analyze reliability data.

Please note that I will not emphasis topics beyond reliability engineering such as its physics, reliability block diagrams, Failure Modes and Effects Analysis, etc.

RELIABILITY ENGINEERING VS SURVIVAL ANALYSIS

Reliability engineering shares similar science with survival analysis since they both analyze the duration of time until one or more events occur, but their terminology are different. Engineers can control experiments with cars, airplanes, laptops, etc. Biostatisticians cannot control their patients. You can't randomize a person to smoke or not to take their heart medication. Engineers can accelerate their valuation of a product or system by pushing test conditions to the absolute limit. For example, they run a car with one quart of oil to induce quick failure and then extrapolate the failure time back to normal (5 quarts of oil) condition Products and systems can break down, be fixed, and then be put back in service only to break again. The key thing to consider is whether the product or system repairable or not. For example, a lightbulb is not a repairable system whereas a car engine is a repairable system.

INTRODUCTION TO PROC RELIABILITY

PROC RELIABILITY can be used to analyze reliability and survival data analysis and for recurrent events data analysis. It can be used to:

- Construct probability plots and fitted life distributions with censored data (left, right, or interval)
- Fit regression model and accelerated life test models with censored data
- Analyze recurrence data from repairable systems.

PROC RELIABILITY offers several probability distributions to estimate failure time such as exponential, extreme value, log-logistic, lognormal, normal, Poisson and Weibull. The exponential distribution assumes constant failure rate but rarely used in engineering. Lognormal and Weibull distribution are commonly used in engineering because of the flexibility of the shape of the distribution. Weibull distribution is the most commonly used probability distribution for modeling reliability data. It is used among engineers and quality practitioners. It can also model hazard functions that are decreasing, increasing or constant, allowing it to describe an item's lifetime to failure. Lognormal distribution is often used to model times to repair a maintainable system

Usually, you want to find the distribution that gives you the best fit of the data. You generate the probability plot to estimate the fitted distribution. Afterward, you test the data into the goodness-of-fit whether the fitted distribution follows a specific probability distribution or not.

A SIMPLE PROC RELIABILITY

This example illustrates the use of PROC RELIABILITY to model life data from a single population using uncensored lifetimes. The following statements create a SAS data set containing complete and right-censored lifetimes of 20 switches. (O'Connor; 2011, pg.103). Some of the rigs had not failed at the time the data collection ended, so the non-failed switches have right-censored lifetimes. The variable *cycles* represents number of cycles of operations. The *censor* variable represents whether a unit fails or not, so 0 if the unit fails and 1 if it doesn't.

```
data rig;
  input cycles censor @@;
datalines;
420 0 890 0 1090 0 1120 0 1400 0 1810 0 1815 0 2150 0 2500 0 2510 0
3030  0 3290 0 3330 0 3710 0 4250 0 5000 1 5000 1 5000 1 5000 1 5000 1
;
run;

proc reliability data = rig;
  distribution Weibull;
  probplot cycles * censor(1);
run;
```

The DISTRIBUTION statement specifies the Weibull distribution for probability plot and maximum likelihood (ML) parameter estimation. The PROBPLOT statement performs the estimation and generates the probability plot. If the DISTRIBUTION statement is not specified, then SAS specifies the normal distribution by default.

The PROBPLOT statement produces the probability plot for a time-to-failure variable using a specified probability distribution. We include the censor variable to indicate which observations are right-censored.

Weibull Parameter Estimates

Parameter	Estimate	Standard Error	Asymptotic Normal 95% Confidence Limits Lower	Upper
EV Location	8.2373	0.1744	7.8954	8.5792
EV Scale	0.6755	0.1492	0.4381	1.0415
Weibull Scale	3779.3260	659.2813	2684.8945	5319.8758
Weibull Shape	1.4803	0.3270	0.9601	2.2825

Figure 1. Weibull Parameter Estimates for rig



Display 2. Probability plot for rig

Figure 1 shows the table of the parameter estimates of scale and shape for Weibull and extreme value distribution. To get the extreme value estimate, you take the natural log of Weibull's estimate.

Display 1 provides the Weibull probability plot as it shows failure times associated with 20 rigs. The x-axis represents cycles. The y-axis represents unreliability.

ACCELERATED LIFE TESTS

Testing products may take longer to record the time in which failures occur. To reduce the actual time of the test, accelerated life testing accelerates the performance by increasing stress levels so you can extrapolate the results back to normal conditions. The goal of accelerated life testing is to record times-to-failure data while in extreme conditions.

For example, it may take years for a laptop to fail under normal conditions. Instead, we could accelerate the performance by increasing the voltage and then extrapolate the failure time back to normal.

You can use SAS to perform statistical analysis of accelerated test data. The following statements create a SAS data set containing 24 electronic parts where they have been subjected to accelerated testing. (O'Connor; 2011, pg.337). The electronic parts are usually operate at 40°C, however, in this example we set the temperature higher. The first eight samples are set to 60°C, other eight samples are 80°C, and the remaining samples are 100°C. The variable *time* represents time of failure in hours. The variable *status* depends on whether a certain part failed or suspended. If a part is active by 250 hours, the status will set to "suspended." A numeric value must be specified for the variable, CENSOR, of PROC RELIABILITY.

```
* Accelerated life testing example ;
data product;
  input temperature time status :$9. @@;
  if status = "Fail" then
    censor = 0;
  else
    censor = 1;
datalines;
60 68 Fail  60 127 Fail  60 186 Fail
60 205 Fail  60 250 Suspended  60 250 Suspended
60 250 Suspended  60 250 Suspended  80 55 Fail
80 63 Fail  80 80 Fail  80 126 Fail
80 137 Fail  80 192 Fail  80 240 Fail
```

```
80 250 Suspended  100 13 Fail  100 15 Fail
100 30 Fail  100 31 Fail  100 47 Fail
100 73 Fail  100 95 Fail  100 98 Fail
;
run;

proc reliability data = product;
  distribution weibull;
  RELATIONPLOT time * censor(1) = temperature /
    fit = model
    plotdata
    plotfit 10 50 90
    relation = arr
    noconf
    slower = 30
    supper = 120
    sref(intersect) = 40
    sreflabel = "Normal Use";
  model time * censor(1) = temperature / relation = arr;
run;
```

The DISTRIBUTION statement will estimate parameter estimates for each group as well as percentile estimates.

The RELATIONPLOT statement produces life-stress relation plots. A life-stress relation plot is useful for analyzing accelerated life tests. The plot displays a relationship between temperature and time. The PLOTDATA option produces the relation plot with failure times. The PLOTFIT specifies the *p*th percentiles of the fitted relationship on the relation plot (50^{th} percentile is the default). Suppose we are interested in the 10^{th}, 50^{th}, and 90^{th} percentiles of failure time at 40*C. The SLOWER and SUPPER options specify lower and upper limits on the stress (temperature) axis scale in the plot. The SREF option draws a vertical line on the plot to indicates the normal use during operation.

The MODEL statement fits a regression model that has an inverse-power stress relationship, called “Arrhenius”. Arrhenius is a popular physics equation that shows the relationship between temperature and life time. Once again, we include the censor variable to indicate which observations are right-censored.

Weibull Percentile Estimates					
			Asymptotic Normal 95% Confidence Limits		
Percent	Estimate	Standard Error	Lower	Upper	Group
0.1	5.86989562	4.55025519	1.28467031	26.8206358	60
0.2	8.8938669	6.2639365	2.23661233	35.3663741	60
0.5	15.4122239	9.44253039	4.6383541	51.211408	60
1	23.3802854	12.7549033	8.02572924	68.110664	60
2	35.5219755	17.0907933	13.8342618	91.2091128	60
5	62.0726942	24.9698028	28.2156603	136.556059	60
10	95.5370647	33.442579	48.1072257	189.728894	60
20	149.762808	46.1774897	81.836419	274.069892	60
30	198.345368	57.794187	112.046323	351.112681	60
40	245.964744	69.8961955	140.924193	429.299288	60
50	295.309706	83.3733895	169.809516	513.562635	60
60	349.049052	99.1849062	199.991326	609.202625	60
70	411.076873	118.860736	233.239742	724.508585	60
80	489.144277	145.633427	272.902723	876.730437	60
90	606.195974	189.480088	328.511818	1118.60073	60
95	709.702108	231.450338	374.523873	1344.8464	60
99	918.212185	323.522623	460.289364	1831.70345	60
99.9	1170.65612	445.862441	554.927318	2469.57704	60

Figure 3. Weibull Percentile Estimates for accelerate life testing at a temperature of 60°C.



Display 2. Percentiles plot for time

Display 2 shows the percentiles plot for time. While the temperature sets to 40, the estimated 10th percentile is 275.98 hours, the estimated 50th percentile is 853.06 hours, and the estimated 90th percentile is 1751.12 hours. To interpret the 10th percentile, that means out of 100 products tested, 10 products will failed and 90 products will continue functioning at 95.37 hours while the temperature sets to 60 temperature.

PREVENTIVE MAINTENANCE

Preventive maintenance (PM) is performed on a regular basis by replacing equipment to avoid future breakdowns and equipment failures. The goal of preventive maintenance is to preserve reliability by replacing worn equipment before they fail. For example, people take their car to the dealership according to the maintenance schedule. Car mechanics replace whatever equipment needs to replace such as oil, oil filter, tires, brakes, etc. The optimal schedule preventive maintenance depends on the time-to-failure distributions of the maintained parts and the failure rate trend of the system. The optimal schedule also depends on the cost of scheduled replacement.

A hazard rate is dependent whether any replacement will increase the probability of failure. If a part has a decreasing hazard rate, then any replacement will increase the probability of failure. If a part has a constant hazard rate, then any replacement will have no effect to the probability of failure. If a part has an increasing hazard rate, then any replacement will decrease the probability of failure.

The assumptions of preventive maintenance require that failure distributions are defined and no other defects are found." We must take into account the effects of the maintenance action on reliability.

The following example of preventive maintenance is the data gives the times between failures of an air conditioner (AC). When an AC fails while in use the cost of replacing the AC is $1000. The cost of replacement during scheduled maintenance is $300. Suppose the AC runs 24/7 and scheduled maintenance take place at least four and a half months. Suppose we want to know the optimal total cost of the preventive maintenance of the AC given three choices: 4 months, 4.6 months and 5 months.

The following statements are the data analysis using PROC RELIABILITY as well as DATA steps and some PROCs.

```
* Preventive Maintenance ;
data ac;
  input time @@;
datalines;
29 12 70 21 29 386 59 27 153 26 326
;
run;

proc reliability data = ac;
  distribution weibull;
```

```
  model time;
  ODS OUTPUT Modprmest = param;
run;

data param;
  set param;
  drop Parm Stderr Lower Upper;
run;

proc transpose data = param out = t_param;
  var Estimate;
run;

data pm (drop = _NAME_) ;
  set t_param (rename = (COL1 = Location COL2 = Scale COL3 = Shape));
  cost_unef = 1000;  * Cost of unexpected failure and replacement;
  cost_sch = 300;  * Cost of scheduled replacement;
  array month{3} (4 4.6 5);
  array total_cost{3};
  do i = 1 to 3;
    sch_main_cost = (cost_unef * cost_sch) / month{i};  * Scheduled maintenance cost;

    * Probability of failure at time (month) ;
    * According to the CDF of Weibull dist. ;
    R = 1 - exp(-((month{i} - Location)/ Scale)**Shape);

    * R has to be between 0 and 1 ;
    if R = . then
      R = 0;

    E_FC = cost_unef * R;     * Expected failure cost ;
    total_cost{i} = sch_main_cost + (cost_unef / month{i}) * E_FC;  * total cost ;
  end;
run;

proc print data = pm noobs label;
  var total_cost1 total_cost2 total_cost3;
  format total_cost1 total_cost2 total_cost3 dollar12.2;
  label total_cost1 = "Total Cost ($) @ 4 Months"
        total_cost2 = "Total Cost ($) @ 4.6 Months"
              total_cost3 = "Total Cost ($) @ 5 Months";
run;
```

Weibull Parameter Estimates				
			Asymptotic Normal 95% Confidence Limits	
Parameter	Estimate	Standard Error	Lower	Upper
Intercept	4.5696	0.3620	3.8600	5.2792
EV Scale	1.1289	0.2524	0.7283	1.7498
Weibull Shape	0.8858	0.1981	0.5715	1.3730

Figure 3. Weibull Parameter Estimates for preventive maintenance.

Total Cost ($) @ 4 Months	Total Cost ($) @ 4.6 Months	Total Cost ($) @ 5 Months
$75,000.00	$73,887.14	$129,329.69

Display 3. Total cost of the preventive maintenance of the AC.

In the data analysis, we specify the probability distribution as 3-parameter Weibull, in which we use to calculate the reliability. There are three parameters: location (or threshold), scale and shape. The location parameter says that we don't record failures until after 4.5696 months. Based on Display 3, the optimal total cost of the preventive maintenance is 4.6 months. Thus, we recommend scheduled maintenance take place at 4.6 months, or 139 days.

CONCLUSION

Reliability engineering has a combination of knowledge between engineering and statistics. The most important priority for reliability engineers is to reduce the likelihood of failures of the system. From a customer's perspective, we want to use a product as long as we desired. From a company's perspective, we would want to save the reputation of a product. In general, the goal of any company is to make money. System that doesn't have a very strong reliability could ruin the business. That is why today's technology needs to have a very long durability so we can rely on them in every day usage.

This paper should give you an idea how reliability engineering works in the industry and how engineers use PROC RELIABILITY to analyze reliability data. Because there are several areas that I haven't touched on, I encourage you to continue researching areas in reliability engineering.

REFERENCES

O'Connor, Patrick, and Kleyner, Andre. 2011. *Practical Reliability Engineering (5)*. 103, 354, 337, 414-415. Hoboken, GB: Wiley.

Gutierrez, Roberto. 2013. "Good as New or Bad as Old? Analyzing Recurring Failures with the RELIABILITY Procedure". *Proceedings of the SAS Global 2013 Conference.* Cary, NC: SAS Institute Inc.
Available at https://support.sas.com/resources/papers/proceedings13/424-2013.pdf

Gutierrez, Roberto. 2014. "Accelerated Tests as an Effective Means of Quality Improvement". Cary, NC: SAS Institute Inc. Available at https://support.sas.com/resources/papers/proceedings14/SAS154-2014.pdf

SAS Institute Inc. (2015), SAS/QC® 14.1 User's Guide The RELIABILITY Procedure, Cary, NC: SAS Institute Inc.

ACKNOWLEDGMENTS

I would like to thank Bobby Gutierrez for his advices and ideas on how to present a paper on reliability engineering to non-engineers at the SAS conferences. I also would like to thank Daniel Waldorf for giving me advices on reliability engineering as well and teaching the course at Cal Poly. Lastly, I would like to thank Kirk Lafler for his feedbacks on this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Danny Rithy
E-mail: rithy.danny@gmail.com
Web: linkedin.com/in/drithy